
INTERNATIONAL ELECTROTECHNICAL COMMISSION 

____________ 

 
FUNCTION BLOCKS (FB) FOR PROCESS CONTROL AND  

ELECTRONIC DEVICE DESCRIPTION LANGUAGE (EDDL) –  
 

Part 4: EDD interpretation 
 

FOREWORD 

1) The International Electrotechnical Commission (IEC) is a worldwide organization for standardization comprising 
all national electrotechnical committees (IEC National Committees). The object of IEC is to promote 
international co-operation on all questions concerning standardization in the electrical and electronic fields. To 
this end and in addition to other activities, IEC publishes International Standards, Technical Specifications, 
Technical Reports, Publicly Available Specifications (PAS) and Guides (hereafter referred to as “IEC 
Publication(s)”). Their preparation is entrusted to technical committees; any IEC National Committee interested 
in the subject dealt with may participate in this preparatory work. International, governmental and non-
governmental organizations liaising with the IEC also participate in this preparation. IEC collaborates closely 
with the International Organization for Standardization (ISO) in accordance with conditions determined by 
agreement between the two organizations. 

2) The formal decisions or agreements of IEC on technical matters express, as nearly as possible, an international 
consensus of opinion on the relevant subjects since each technical committee has representation from all 
interested IEC National Committees.  

3) IEC Publications have the form of recommendations for international use and are accepted by IEC National 
Committees in that sense. While all reasonable efforts are made to ensure that the technical content of IEC 
Publications is accurate, IEC cannot be held responsible for the way in which they are used or for any 
misinterpretation by any end user. 

4) In order to promote international uniformity, IEC National Committees undertake to apply IEC Publications 
transparently to the maximum extent possible in their national and regional publications. Any divergence 
between any IEC Publication and the corresponding national or regional publication shall be clearly indicated in 
the latter. 

5) IEC itself does not provide any attestation of conformity. Independent certification bodies provide conformity 
assessment services and, in some areas, access to IEC marks of conformity. IEC is not responsible for any 
services carried out by independent certification bodies. 

6) All users should ensure that they have the latest edition of this publication. 

7) No liability shall attach to IEC or its directors, employees, servants or agents including individual experts and 
members of its technical committees and IEC National Committees for any personal injury, property damage or 
other damage of any nature whatsoever, whether direct or indirect, or for costs (including legal fees) and 
expenses arising out of the publication, use of, or reliance upon, this IEC Publication or any other IEC 
Publications.  

8) Attention is drawn to the Normative references cited in this publication. Use of the referenced publications is 
indispensable for the correct application of this publication. 

9) Attention is drawn to the possibility that some of the elements of this IEC Publication may be the subject of 
patent rights. IEC shall not be held responsible for identifying any or all such patent rights. 

International Standard IEC 61804-4 has been prepared by subcommittee 65E: Devices and 
integration in enterprise systems, of IEC technical committee 65: Industrial-process 
measurement, control and automation. 

This first edition cancels and replaces IEC TR 61804-4 published in 2006. This edition 
constitutes a technical revision. 

This edition includes the following significant technical changes with respect to the previous 
edition: 

• New paragraph:  
– EDDL data description 
– EDDL METHOD programming and usage of builtins 
– Edit session 
– Offline and online configuration 

– EDDL communication description  

• Enhancements in paragraph EDDL user interface descriptions 

The text of this standard is based on the following documents: 

FDIS Report on voting 
65E/465/FDIS 65E/481/RVD 

 
Full information on the voting for the approval of this standard can be found in the report on 
voting indicated in the above table. 

This publication has been drafted in accordance with the ISO/IEC Directives, Part 2. 

A list of all parts in the IEC 61804 series, published under the general title Function blocks 
(FB) for process control and electronic device description language (EDDL), can be found on 
the IEC website. 

Future standards in this series will carry the new general title as cited above. Titles of existing 
standards in this series will be updated at the time of the next edition. 

The committee has decided that the contents of this publication will remain unchanged until 
the stability date indicated on IEC web site under "http://webstore.iec.ch" in the data related 
to the specific publication. At this date, the publication will be  

• reconfirmed, 

• withdrawn, 

• replaced by a revised edition, or 

• amended. 

 

IMPORTANT – The 'colour inside' logo on the cover page of this publication indicates 
that it contains colours which are considered to be useful for the correct 
understanding of its contents. Users should therefore print this document using a 
colour printer. 

 

  

 
 
 
 
 
 
 
 
 
 
 
 
 
 

 

Full document details: IEC 61804-4 2015

http://www.avadocuments.com/doc/iec-61804-4-2015
http://www.avadocuments.com/doc/iec-61804-4-2015


 – 10 – IEC 61804-4:2015  IEC 2015 

INTRODUCTION 

This part of IEC 61804 was developed using material from FDI Cooperation LLC (Foundation™ 
Fieldbus1, HART®2 Communication Foundation (HCF), PROFIBUS™3 Nutzerorganisation e.V. 
(PNO)), OPC Foundation (OPCF) and FDT Group. IEC 61804 has the general title "Function 
blocks (FB) for process control and Electronic Device Description Language (EDDL)".  

This editon does reflect many of the various rules defined by the different communication 
foundations, however it is not a complete representation of those rules defined by each of the 
communication foundations today.  Therefore, an EDD application and EDD developer will 
need to rely on both IEC 61804-4 and the respective communication foundation documents 
(e.g. specifications, test requirements, test cases) to develop a conformant application that 
will meet foundation registration requirements. 

Conformity assessment of an EDD application is the responsibility of the respective 
communication foundations. In cases of any ambiguity, the rules of the respective 
communication foundations apply. 

This part of IEC 61804 

• contains an overview of the use of EDDL; 

• provides examples demonstrating the use of the EDDL constructs; 

• shows how the use cases are fulfilled; and  

• shows the proper EDD application interpretation for each example. 

This part of IEC 61804 is not an EDDL tutorial and is not intended to replace the EDDL 
specification. 

Instructions are provided for the EDD application, which describe what will be performed 
without prescribing the technology used in the host implementation. For example, the FILE 
construct describes data that is stored by the EDD application on behalf of the EDD. The FILE 
construct does not specify how the data is stored. The EDD application can use a database, a 
flat file, or any other implementation it chooses. 

  

___________ 
1  FOUNDATION™ Fieldbus is the trademark of the Fieldbus Foundation. This information is given for the 

convenience of users of this document and does not constitute an endorsement by IEC of the product named. 
Equivalent products may be used if they can be shown to lead to the same results. 

2  HART® is the registered trademark of the HART Communication Foundation. This information is given for the 
convenience of users of this document and does not constitute an endorsement by IEC of the product named. 
Equivalent products may be used if they can be shown to lead to the same results. 

3  PROFIBUS and PROFINET are the trademarks of the PROFIBUS Nutzerorganisation e.V. This information is 
given for the convenience of users of this document and does not constitute an endorsement by IEC of the 
product named. Equivalent products may be used if they can be shown to lead to the same results. 

FUNCTION BLOCKS (FB) FOR PROCESS CONTROL AND  
ELECTRONIC DEVICE DESCRIPTION LANGUAGE (EDDL) –  

 
Part 4: EDD interpretation 

 
 
 

1 Scope 

This part of IEC 61804 specifies EDD interpretation for EDD applications and EDDs to support 
EDD interoperability. This document is intended to ensure that field device developers use the 
EDDL constructs consistently and that the EDD applications have the same interpretations of 
the EDD. It supplements the EDDL specification to promote EDDL application interoperability 
and improve EDD portability between EDDL applications.  

2 Normative references 

The following documents, in whole or in part, are normatively referenced in this document and 
are indispensable for its application. For dated references, only the edition cited applies. For 
undated references, the latest edition of the referenced document (including any 
amendments) applies. 

IEC 61784-1, Industrial communication networks – Profiles – Part 1: Fieldbus profiles 

IEC 61784-2, Industrial communication networks – Profiles – Part 2: Additional fieldbus 
profiles for real-time networks based on ISO/IEC 8802-3 

IEC 61804-2, Function blocks (FB) for process control – Part 2: Specification of FB concept 

IEC 61804-34, Function blocks (FB) for process control and Electronic device description 
language (EDDL) − Part 3: EDDL syntax and semantics 

IEC 61804-55, Function blocks (FB) for process control and Electronic device description 
language (EDDL) − Part 5: EDDL Builtin library 

ISO/IEC 10918 (all parts), Information technology – Digital compression and coding of 
continuous-tone still images 

ISO/IEC 15948, Information technology – Computer graphics and image processing – 
Portable Network Graphics (PNG): Functional specification 

3 Terms, definitions, abbreviated terms, acronyms and conventions 

For the purposes of this document, the terms and definitions given in IEC 61804-3 as well as 
the following apply. 

___________ 
4  To be published. 

5  To be published. 

3.1 General terms and definitions 

3.1.1  
EDD developer 
individual or team that develops an EDD 

3.1.2  
container 
user interface elements that contain other user interface elements 

Note 1 to entry: Containers can include menus, windows, dialogs, tables, pages, groups, and other containers. 

3.1.3  
contained item 
user interface elements that can be contained in containers 

Note 1 to entry: Contained items can include variables, methods, graphs, charts, images, static text. 

3.1.4  
device developer 
individual or team that develops a device and an EDD that describes the device 

3.1.5  
handheld 
device with limited display resolution that restricts EDD applications user interface 

3.2 Terms and definitions related to modular devices 

3.2.1  
channel 
connection to a process that is being measured or controlled  

3.2.2  
component 
software or hardware item contained within the modular device concept 

Note 1 to entry: A component cannot function separately from a modular device hosting it. A component may 
support one or more types of modular devices. 

3.2.3  
interface 
basic declarations of basic constructs  

Note 1 to entry: An interface defines all public parts that components may use. 

3.2.4  
modal window 
a child window that requires users to interact with it before they can return to operating the 
parent application, thus preventing the workflow on the application main window 

3.2.5  
modular device 
device that can contain a variety of software and or hardware components 

3.3 Abbreviated terms and acronyms 
 
CP Communication Profile 

CPF Communication Profile Family 

EDD Electronic Device Description 

EDDL Electronic Device Description Language 

PC Personal computer 

PI PROFIBUS and PROFINET International 

PI PROFILE PA PI specific profile for Process Automation field devices 

3.4 Conventions 

There exists some differences using and interpreting EDDL based on the used communication 
network and device model. The different communication networks used within this part of 
IEC 61804 are: 

• HART (according to IEC 61784-1 CPF 9) 

• FOUNDATION fieldbus (according to IEC 61784-1 CPF 1) 

• PROFIBUS (according to IEC 61784-1 CPF 3) 

• PROFINET (according to IEC 61784-2 CPF 3) 

EDD examples in this standard show parts of EDD implementations and are incomplete. 

EDDL keywords are written in upper case letters. If more than one basic construct item is 
meant the keyword is written in uppercase letters added with a lower case ‘s’ for example 
VARIABLEs. 

The capitalized word Builtin refers to functions specified in IEC 61804-5. 

EXAMPLE Builtin MenuDisplay refers to the Builtin named MenuDisplay specified in IEC 61804-5. 

4 EDDL user interface description 

4.1 Overview 

Most EDD applications can be characterized as either a PC application or a handheld 
application. Due to the relatively small screen of a handheld device, handheld applications 
can only display a small amount of information at any given time. On the other hand, PC 
applications can provide a much more beneficial user interface, largely due to their larger 
screen size. 

To support the capabilities of PC applications, the MENU construct has been extended in 
IEC 61804-3 compared to previous definitions in IEC 61804-2. Due to the differences in the 
user interfaces of PC applications and handheld applications, it is expected that many devices 
will define two MENU hierarchies – one for handheld applications and the other for PC 
applications. Some MENUs may be used in both hierarchies. Therefore, the entire hierarchy 
does not need to be specified twice. 

Different menu structures for different classes of applications are possible. This standard shall 
be used to create menu structures in an EDD that are interpreted by applications in an 
unambiguous way. To provide interoperability across applications, this standard shall be 
followed. 

4.2 Menu conventions for handheld applications 

EDD applications use specific menus in the EDDs to show the user interface of the device 
(see Table 1). In addition user interface items can be described for handhelds and PCs at 
once. For handhelds strings can have beside a language code, a specific country code zz to 
specify shorter strings or lower resolution images (see IEC 61804-3). 

Table 1 – List of defined root menu identifiers for handhelds 

Menu identifier Default STYLE Short description 

root_menu TABLE Handheld root menu for HART devices. 
This is mandatory for all HART EDDs 

Menu_Top* MENU Handheld root menu prefix for BLOCK_A 
MENU_ITEMs for FOUNDATION fieldbus 
devices e.g. Menu_Top_TB 

 
4.3 Menu conventions for PC-based applications 

4.3.1 Overview 

EDD applications use special menus in the EDDs to show the user interface of the device. 
Such menus are defined for diagnostic, process variables, device features, and offline 
configuration. Table 2 defines identifiers for the different root menus with its default STYLEs. 
The default STYLE is used by the EDD application if the STYLE attribute is not defined in the 
MENU. The “Usage” column in Table 2 defines how the EDD application has online or offline 
access to the device parameter (see Clause 8).  

Table 2 – List of defined root menu identifiers for PC-based devices 

MENU identifier Default STYLE Usage Short description 

device_root_menu MENU Online Device feature views for set up 

diagnostic_root_menu MENU Online Diagnostic views 

maintenance_root_menu MENU Online Maintenance feature views 

offline_root_menu TABLE Offline Offline configuration 

process_variables_root_menu MENU Online Process variable views 

 
For HART EDDs the EDD application shall display offline_root_menu with a default STYLE 
WINDOW if STYLE is not defined. 

PROFIBUS and PROFINET allow in submenus to define different parameter access by 
defining ACCESS ONLINE or ACCESS OFFLINE.  

The EDDs should contain the online root menus from Table 2. The online root menus are 
optional for PROFIBUS, PROFINET and FOUNDATION fieldbus. The offline root menu is 
optional for FOUNDATION fieldbus. HART EDDs shall have all root menus in Table 2. 

4.3.2 Online Root Menus 

4.3.2.1 General 

If none of the online root menus exists then communication profile specific entry points shall 
be used (see Table 3).  

Table 3 – Fall back alternatives for online root menus 

Communication profile Description of fall back alternatives 

FOUNDATION fieldbus Menus declared in the BLOCK_A MENU_ITEMS attribute, e.g. 
device_root_menu_aiblock. If block based menus are not defined, then a MENU of 
style TABLE shall be generated from the BLOCK_A PARAMETERS attribute. 

HART root_menu 

PROFIBUS, PROFINET Menu_Main_Specialist if exists or Menu_Main_Maintenance. These menus include 
online and offline sub-menus. 

 

4.3.2.2 Diagnostic Root Menu 

The diagnostic_root_menu includes views that show the device state, detailed diagnostic 
information and may include graphical views that show, for example, a valve signature. 

4.3.2.3 Process variable Root Menu 

The process_variable_root_menu includes views that show process measurements and set 
points with their quality and important information for process operators, for example, ranges.  

4.3.2.4 Device Root Menu 

The device_root_menu includes features for a device. These features can be split into 
process-related and device-specific features. This structure is not required if the number of 
features is too small for splitting. Submenus that represent any structuring are allowed on the 
device feature menu. In case of such submenus, the menus that are underneath can be split 
into process-related and device-specific features. 

4.3.2.5 Maintenance Root Menu 

The maintenance_root_menu should contain features for maintaining the device during the 
runtime phase and e.g. showing information about last maintenance inspection. 

4.3.3 Offline Root Menu 

The offline_root_menu is a menu hierarchy including e.g. data items and methods for offline 
configuration. It contains, in particular, all of the application-specific parameters of the device 
and may also contain important read-only and writeable variables. For more information about 
application-specific parameters, see 9.3. The menu can have offline methods, for example, 
configuration assistants. 

If the offline root menu does not exist, the communication profile specific entry point shall be 
used (see Table 4). 

Table 4 – Fall back alternatives for offline root menus 

Communication profile Description of fall back alternatives 

FOUNDATION fieldbus BLOCK_A PARAMETERS 

HART MENU upload_variables 

PROFIBUS, PROFINET Table_Main_Specialist if exists or Table_Main_Maintenance 

 
4.3.4 Example of EDD menu structure 

The EDD example in Figure 1 includes additional menus that can be added to an EDD for PC 
applications. These menus are additional to the existing menus for the applications. This 
example is specific for a device with an interface according to HART. Examples for devices 
with an interface according FOUNDATION fieldbus, PROFIBUS and PROFINET would be very 
similar. 

MENU diagnostic_root_menu 
{ 
    LABEL "Diagnostics"; 
    STYLE MENU;   /* not required to define STYLE in this case, */ 
      /* because of default STYLE of the root menu */ 
    ITEMS 
    { 
        status_window,   /* menu: style=window */  
  self_test   /* method */ 
    } 
} 
 
MENU status_window 
{ 
    LABEL "Status"; 
    STYLE WINDOW; 
    ITEMS 
    { 
        standard_diagnostics_page,  /* menu: style=page */ 
        devspec_diagnostics_page  /* menu: style=page */ 
    } 
} 
 
MENU standard_diagnostics_page 
{ 
    LABEL "Standard"; 
    STYLE PAGE; 
    ITEMS 
    { 
         device_status    /* variable */ 
    } 
} 
 
MENU devspec_diagnostics_page 
{ 
    LABEL "Device Specific"; 
    STYLE PAGE; 
    ITEMS 
    { 
        xmtr_specific_status_1,  /* variable */ 
        xmtr_specific_status_2  /* variable */ 
    } 
} 
 
METHOD self_test 
{ 
    LABEL "Self Test"; 
    DEFINITION 
    { 
         /* elided */ 
    } 
} 
 
MENU maintenance_root_menu 
{ 
    LABEL "Maintenance"; 
    STYLE MENU;   /* not required to define STYLE in this case, */ 
      /* because of default STYLE of the root menu */ 
 
    ITEMS 
    { 
        device_mode_dialog,   /* menu: style=dialog */ 
        teach_in    /* method */ 
    } 
} 
MENU device_mode_dialog 
{ 
    LABEL "Device Mode"; 
    STYLE DIALOG; 
    ITEMS 
    { 
        mode_page                   /* menu: style=page */ 
    } 
} 
 
MENU mode_page 
{ 
    LABEL "Process Variables"; 
    STYLE PAGE; 

    ITEMS 
    { 
        transducer_group,           /* menu: style=group */ 
        function_group              /* menu: style=group */ 
    }    
} 
 
MENU transducer_group 
{ 
    LABEL "Transducer"; 
    STYLE GROUP; 
    ITEMS 
    { 
        trans_target_mode,          /* variable */ 
        trans_actual_mode           /* variable */ 
    } 
} 
 
MENU function_group 
{ 
    LABEL "Function"; 
    STYLE GROUP; 
    ITEMS 
    { 
        func_target_mode,           /* variable */ 
        func_actual_mode            /* variable */ 
    } 
} 
 
METHOD teach_in 
{ 
    LABEL "Teach-in"; 
    DEFINITION 
    { 
        /* elided */ 
    } 
} 
 
MENU process_variables_root_menu 
{ 
    LABEL "Process Variables"; 
    STYLE MENU;   /* not required to define STYLE in this case, */ 
      /* because of default STYLE of the root menu */ 
 
    ITEMS 
    { 
        overview_window,  /* menu: style=window */ 
        primary_vars_window  /* menu: style=window */ 
    } 
} 
 
MENU overview_window 
{ 
    LABEL "Overview"; 
    STYLE WINDOW; 
    ITEMS 
    { 
        process_vars_page  /* menu: style=page */ 
    } 
} 
 
MENU process_vars_page 
{ 
    LABEL "Process Variables"; 
    STYLE PAGE; 
    ITEMS 
    { 
        pressure_group,  /* menu: style=group */ 
        temperature_group  /* menu: style=group */ 
    }  
} 
 
MENU pressure_group 
{ 
    LABEL "Pressure"; 
    STYLE GROUP; 
    ITEMS 
    { 
        pv_digital_value,  /* variable */ 

        pv_upper_range_value,  /* variable */ 
        pv_lower_range_value  /* variable */ 
    } 
} 
 
MENU temperature_group 
{ 
    LABEL "Temperature"; 
    STYLE GROUP; 
    ITEMS 
    { 
        sv_digital_value,  /* variable */ 
        sv_upper_range_value,  /* variable */ 
        sv_lower_range_value  /* variable */ 
    } 
} 
 
MENU primary_vars_window 
{ 
    LABEL "Primary Variables"; 
    STYLE WINDOW; 
    ITEMS 
    { 
        pressure_chart_page,  /* menu: style=page */ 
        temperature_chart_page /* menu: style=page */ 
    } 
} 
 
MENU pressure_chart_page 
{ 
    LABEL "Pressure"; 
    STYLE PAGE; 
    ITEMS 
    { 
        pressure_chart   /* chart */ 
    } 
} 
 
CHART pressure_chart 
{ 
    /* elided */ 
} 
 
MENU temperature_chart_page 
{ 
    LABEL "Temperature"; 
    STYLE PAGE; 
    ITEMS 
    { 
        temperature_chart  /* chart */ 
    } 
} 
 
CHART temperature_chart 
{ 
    /* elided */ 
} 
 
MENU device_root_menu 
{ 
    LABEL "Device"; 
    STYLE MENU;   /* not required to define STYLE in this case, */ 
      /* because of default STYLE of the root menu */ 
 
    ITEMS 
    { 
        process_related_window, /* menu: style=window */ 
        device_specific_window, /* menu: style=window */ 
        master_reset   /* method */ 
    } 
} 
 
MENU process_related_window 
{ 
    LABEL "Process Related"; 
    STYLE WINDOW; 
    ITEMS 
    { 
        identification_page,  /* menu: style=page */ 

        output_info_page  /* menu: style=page */ 
    } 
} 
 
MENU identification_page 
{ 
    LABEL "Identification"; 
    STYLE PAGE; 
    ITEMS 
    { 
        tag,    /* variable */ 
        manufacturer,   /* variable */ 
        device_type,   /* variable */ 
        device_revision,  /* variable */ 
        descriptor,   /* variable */ 
        message   /* variable */ 
    } 
} 
 
 
MENU output_info_page 
{ 
    LABEL "Output Information"; 
    STYLE PAGE; 
    ITEMS 
    { 
        range_values_group,  /* menu: style=group */ 
        sensor_limits_group  /* menu: style=group */ 
    } 
} 
 
MENU range_values_group 
{ 
    LABEL "Range Values"; 
    STYLE GROUP; 
    ITEMS 
    { 
        pv_units,   /* variable */ 
        pv_urv,   /* variable */ 
        pv_lrv    /* variable */ 
    } 
} 
 
MENU sensor_limits_group 
{ 
    LABEL "Sensor Limits"; 
    STYLE GROUP; 
    ITEMS 
    { 
        sensor_units,   /* variable */ 
        upper_sensor_limit,  /* variable */ 
        lower_sensor_limit  /* variable */ 
    } 
} 
 
MENU device_specific_window 
{ 
    LABEL "Device Specific"; 
    STYLE WINDOW; 
    ITEMS 
    { 
        identification_page,  /* menu: style=page */ 
        calibration_page  /* menu: style=page */ 
    } 
} 
 
MENU calibration_page 
{ 
    LABEL "Calibration"; 
    STYLE PAGE; 
    ITEMS 
    { 
        sensor_limits_group,  /* menu: style=group */ 
        sensor_trim_group  /* menu: style=group */ 
    } 
} 
 
MENU sensor_trim_group 
{ 

    LABEL "Sensor Trim"; 
    STYLE GROUP; 
    ITEMS 
    { 
         upper_sensor_trim_point, /* variable */ 
         lower_sensor_trim_point, /* variable */ 
         sensor_trim   /* method */ 
    } 
} 
 
METHOD master_reset 
{ 
    LABEL "Master Reset"; 
    DEFINITION 
    { 
        /* elided */ 
    } 
} 

Figure 1 – EDD example of root menus 

4.3.5 User interface 

4.3.5.1 Diagnostics 

Figure 2 shows an example of an EDD application for diagnostics. 

 

Figure 2 – Example of an EDD application for diagnostics 

4.3.5.2 Process variables 

Figure 3 and Figure 4 show examples of EDD applications for process variables. 

 

Figure 3 – Example of an EDD application for process variables 

 

Figure 4 – Example of an EDD application for primary variables 

4.3.5.3 Device features 

Figure 5, Figure 6 and Figure 7 show examples of EDD applications for device features. 

 

Figure 5 – Example of an EDD application for process-related device features 

 

 

Figure 6 – Example of an EDD application for device features 

Figure 7 – Example of an EDD application for maintenance features 

4.4 Containers and contained items 

4.4.1 Overview 

The user interface extensions are based on a simple user interface model. The model 
consists of two concepts: 

• containers; and 

• contained items. 

Containers are so named because they contain other user interface elements. Containers may 
include: menus, windows, dialogs, tables, pages, groups, and other containers. Containers 
correspond to a MENU. They are distinguished from one another via the STYLE attribute. This 
STYLE attribute indicates how the MENU will be displayed.  

Contained items include e.g. variables, methods, edit displays, graphs, charts, images, static 
text.  

4.4.2 Containers 

4.4.2.1 Permitted and default STYLEs 

Table 5 defines permitted user interface items in containers, substitutes and default STYLEs 
are not defined. If the style of a menu is not defined, then a default style may be used 
depending on the menu where the menu is contained and the content of the menu. 

The default style of a menu that is not contained in a menu has the style TABLE. The HART 
root_menu is shown as a parameter table. 

General rules: 

• If the ACCESS online or offline of a contained MENU is different to the access of the 
container, the EDD application shall use STYLE WINDOW in a window otherwise shall use 
STYLE DIALOG. 

Table 5 – Permitted contained items and default STYLES 

Container Permitted contained items Default STYLE of 
contained items, used if 

STYLE is not defined 

MENU X X D D 2 X 7 2 2 2 2 X X 6 6 MENU of STYLE MENU, if 
no data item is in the menu. 

DIALOG, if any data item is 
in the menu. 

DIALOG, 
WINDOW 

X 4 X X X 4 X X X X X 4 4 X 8 PAGE 

PAGE X 4 D X X 4 X X X X X 4 4 X 3 GROUP 

GROUP X 4 D 1 D 4 X 1 X X X 4 4 X 3 GROUP if the parent of the 
parent GROUP is not of 
STYLE GROUP. 

DIALOG rendered as button 
if the parent of the parent 
GROUP is of STYLE 
GROUP. 

TABLE 4 D D D X 4 X X X 5 X 4 4 6 6 TABLE 
Key 

X Allowed. 

D Default STYLE shall be used (see column “Default STYLE of submenu if not defined”). 

1 Only one sublevel is allowed e.g. GROUP may contain a GROUP but this GROUP shall not contain GROUPs, 
COLLECTIONs or RECORDs. 

2 Shall be rendered in a dialog. 

3 Shall be interpreted as COLUMNBREAK. 

4 Shall be rendered as button or hyperlink. 

5 Shall be rendered as button, hyperlink or inline. 

6 Shall be rendered as a line or ignored. 

7 Shall be rendered as button, hyperlink or ignored. 

8 SEPARATOR shall be treated as COLUMNBREAK. 

 
4.4.2.2 Menu 

A menu of STYLE MENU takes the form of a drop-down menu, a pop-up menu or a navigation 
bar.  

4.4.2.3 Window 

A menu of STYLE WINDOW takes the form of a modeless window.  

4.4.2.4 Dialog 

A MENU of STYLE DIALOG shall be a modal window. If a dialog contains a subordinated 
window, the EDD application shall manage the window as a modal subdialog. 

4.4.2.5 Table 

A menu of STYLE TABLE shall be rendered as a table. Each item shall be represented as one 
or multiple rows, e.g. ARRAYs. The submenus build a hierarchy, which shall be shown.  

The STYLE attribute (STYLE = TABLE) should be defined only on the root menu of a 
hierarchy. 

4.4.2.6 Page 

A menu of STYLE PAGE within a WINDOW or DIALOG shall be rendered as a tab. 

The EDD application shall interpret ROWBREAKs or other items between pages as horizontal 
separations.  

COLUMNBREAKs and SEPARATORs between pages shall be ignored. 

4.4.2.7 Group 

A menu of STYLE GROUP takes the form of a group box. In order to maintain a clear 
arrangement on the user interface, the following restricting rules apply. 

• The nesting level of GROUP declarations is restricted to two. In other words, a GROUP 
may contain further GROUPs but these shall not contain further GROUPs. 
When a second layer GROUP contains a GROUP, this GROUP shall be rendered as either 
a button or a hyperlink. The LABEL shall appear on the button or as the hyperlink text. 
When the button or hyperlink is pressed, a corresponding dialog shall be opened on top of 
the calling dialog or window. 

Within a GROUP, usually logically related parameters and methods are grouped. The title of 
the GROUP indicates this relation, for example, “AnalogInput 1”. Often the same string is also 
used in the labels of the GROUP items, for example, “AnalogInput1_StaticRevision” or 
“Analog Input1_Blockmode”. In order to avoid this duplication of information, the EDD 
developer may assemble the items of a GROUP in a COLLECTION in order to override the 
original labels. If the parameters are to be referred to in a GROUP the parameters can be 
referenced using the COLLECTION; in other cases, especially without additional semantic 
context, the parameters can be referenced directly. 

Figure 8 shows an EDD example with COLLECTION MEMBERS within a MENU of STYLE 
GROUP. 

VARIABLE ai1_strev 
{ 
    LABEL "Analog Input 1: Static Revision"; 
    TYPE INTEGER(4); 
} 
 
VARIABLE ai1_blomo 
{ 
    LABEL "Analog Input 1: Block Mode"; 
    TYPE INTEGER(1); 
} 
 
MENU ai1_group_double    /* Leads to double display of */ 
{ 
    LABEL "Analog Input 1"; 
    STYLE GROUP; 
    ITEMS 
    { 
        ai1_strev, 
        ai1_blomo 
    } 
} 
 
COLLECTION ai1_groupcol 
{ 
    MEMBERS 
    { 
        strev,ai1_strev,"Static Revision"; 
        blomo,ai1_blomo,"Block Mode"; 
    } 
} 
 
MENU ai1_group_single    /* Leads to single display of */ 
{ 
    LABEL "Analog Input 1"; 
    STYLE GROUP; 
    ITEMS 
    { 
        ai1_groupcol.strev, 
        ai1_groupcol.blomo 
    } 
} 

Figure 8 – Usage of COLLECTION MEMBERS in MENUs of STYLE GROUP 

4.4.3 Contained items 

4.4.3.1 Overview 

How a container renders contained items is described in 4.4.3. All containers render 
contained items in the same way. Contained items, such as methods, variables and others, 
have a LABEL attribute. In order not to disturb the layout of the EDD application, for example, 
with oversized buttons for methods, the EDD developer should not use very long strings for 
LABELs. For this reason, some EDD applications may truncate these labels. 

4.4.3.2  Methods 

A method should be rendered as a button or a hyperlink. The LABEL of the corresponding 
METHOD should appear on the button or as the hyperlink text. When the button or hyperlink 
is pressed, the corresponding METHOD should be executed. 

4.4.3.3 Variables 

4.4.3.3.1 General 

The label, value and, if defined, the units of the variable should be displayed in a manner 
consistent with the definition of the corresponding VARIABLE.  

The general handling of variables are as follows. 

• HANDLING = READ or the menu item attribute is READ_ONLY: the value of the variable 
should not be editable. 

• CLASS = DYNAMIC: the value should be updated continuously with current read values 
from the device. 

4.4.3.3.2 Variable of TYPE BIT_ENUMERATED 

Multiple bits can be packaged in a single-bit enumerated variable. Each bit could have a 
distinct meaning. It is possible to display each bit separately and, in addition, to display the 
whole BIT_ENUMERATED variable. BIT_ENUMERATED variables may be displayed as 
checkboxes. 

In case of showing a bit of a BIT_ENUMERATED variable, the variable reference should be 
extended with a mask in square brackets and the LABEL of the variable is not shown. 

Figure 9 shows an EDD example for displaying single bits of a BIT_ENUMERATED variable. 

VARIABLE var1 
{ 
 LABEL “Var 1”; 
 TYPE BIT_ENUMERATED 
 { 
  { 0x1, “Bit 0”}, 
  { 0x2, “Bit 1”}, 
  { 0x4, “Bit 2”} 
} 
} 
 
MENU diagnostic_info 
{ 
 LABEL “Diagnostic”; 
 ITEMS 
 { 
  var1[0x1], // Bit 0 
  var1[0x2], // Bit 1 
  var1[0x4] // Bit 2 
 } 
} 

Figure 9 – Displaying single bits of BIT_ENUMERATED  

Figure 10 shows an EDD example that shows the differences between the full display of a 
BIT_ENUMERATED variable and how it may be displayed if all bits are explicitly addressed. 

VARIABLE var1 
{ 
 LABEL “Var 1”; 
 TYPE BIT_ENUMERATED 
 { 
  { 0x1, “Bit 0”}, 
  { 0x2, “Bit 1”}, 
  { 0x4, “Bit 2”} 
} 
} 
 
MENU var1_group 
{ 
 LABEL var1.LABEL; 
 STYLE GROUP; 
 ITEMS 
 { 
  var1[0x1], // Bit 0 
  var1[0x2], // Bit 1 
  var1[0x4] // Bit 2   
 } 
} 
 
MENU diagnostic_info 
{ 
 LABEL “Diagnostic”; 
 STYLE PAGE; 
 ITEMS 
 { 
  var1,  // all bit should be shown 
   COLUMNBREAK, 
  var1[0x1], // Bit 0 
  var1[0x2], // Bit 1 
  var1[0x4], // Bit 2 
  COLUMNBREAK, 
  var1_group 
 } 
} 

Figure 10 – Displaying multiple bits of BIT_ENUMERATED  

Figure 11 shows how the result of the Figure 10 EDD example in an EDD application may 
appear. 

Diagnostic

Bit 0Var 1
Bit 1
Bit 2

Bit 0
Bit 1
Bit 2

Var 1
Bit 0
Bit 1
Bit 2

 
Figure 11 – Example of an EDD application for a variable of type BIT_ENUMERATED 

4.4.3.3.3 Variable of TYPE INDEX 

When a VARIABLE of type INDEX is presented to the user, the EDD application shall 
determine the text that shall be displayed by following rules: 

1) if the description in the referenced ELEMENTS exist then this shall be displayed else;  
2) if the LABEL of the variable exist then this shall be displayed else: 
3) the LABEL of the array shall be displayed with the numeric value of the INDEX e.g. 

myarray[3]. 

If the INDEX value is out of range of the related array, a text should inform the user that the value is 
out of range. 

 If the variable is editable, it should be presented as a combo box. 

4.4.3.4 Records and collections 

RECORDs and COLLECTIONs shall be displayed as MENU with STYLE GROUP. 
COLLECTIONs can contain COLLECTIONs. This is handled like nested groups (see 4.4.2.7). 
The rules about nesting groups apply also in this case.  

4.4.3.5 Arrays and lists 

Value arrays, reference arrays and LISTs shall be rendered as a scrollable area within the 
container. The item LABEL shall always be visible. 

NOTE HART does not support arrays and lists references in containers. 

4.4.3.6 Edit displays 

An EDIT_DISPLAY shall be rendered in the container as either a button or a hyperlink. The 
LABEL of the corresponding EDIT_DISPLAY shall appear on the button or as the hyperlink 
text.  

When the button or hyperlink is activated, the corresponding EDIT_DISPLAY shall be opened 
as a modal window. The content of an EDIT_DISPLAY contains DISPLAY_ITEMS and 
EDIT_ITEMS. The EDD application shall display items from DISPLAY_ITEMS above of the 
items of EDIT_ITEMS. The  DISPLAY_ITEMS shall be displayed as read only items and 
EDIT_ITEMS shall be handled like ITEMS from a MENU of STYLE DIALOG. 

4.4.3.7 Graphs 

A graph should be displayed in a manner that is consistent with the definition of the 
corresponding GRAPH. Refer to the layout rules defined in 4.5 and IEC 61804-3 for 
information on the affect of the WIDTH attribute on the layout of the GRAPH. 

4.4.3.8 Charts 

A chart should be displayed in a manner that is consistent with the definition of the 
corresponding CHART. Refer to the layout rules defined in 4.5 and IEC 61804-3 for 
information on the affect of the WIDTH attribute on the layout of the CHART. 

4.4.3.9 Images 

The images that are referenced by the MENU should be displayed.  

The EDD application shall display an IMAGE in original size, if it is not referenced with a 
menu item INLINE qualifier. The EDD application shall insert a ROWBREAK above the 
IMAGE, if items exists before the IMAGE and insert a ROWBREAK below the IMAGE, if items 
exists after the IMAGE. 

The EDD application shall scale down an IMAGE with a menu item INLINE qualifier if the 
IMAGE width is larger than the column width in which the IMAGE is contained. 

The EDD application shall not scale up an IMAGE. 

The EDD application shall keep original aspect ratio of IMAGEs. 

4.4.3.10 Static text 

The text that is referenced by the MENU should be displayed. The text will be wrapped in 
multiple lines, if the text is longer than the width of dialog, window, page, group, or column. 

4.4.3.11 Grid 

The LABEL of the grid is shown at first and below the grid area with the width of the dialog, 
window, page, group, or column and the height that is needed to show the data or what is 
available on the screen. Scroll bars are used to scroll the grid area, if the width or height is 
too small to show the complete data. 

4.5 Layout rules 

4.5.1 Overview 

Layout rules are rules that the EDD application shall follow for arranging the content of 
windows and dialogs on the display. 

Each container defines the bounding box of the container. This is the area of the container 
where its contents shall be displayed. For example, the bounding box of a window or dialog is 
the entire window except the border and title bar. 

The ITEMS of a container shall be displayed in the same order they appear in the EDD. The 
contents of the container should be organized starting at the upper left area of the container. 
The items shall be displayed vertically down the container. When a COLUMNBREAK is 
encountered, a new column shall be created. The following items shall be displayed in the 
next column starting at the top of the container. This process continues until all items have 
been displayed. Column breaks shall only be introduced when a COLUMNBREAK is 
encountered. The EDD application shall not introduce columns breaks on its own. 

If static text contains carriage return and line feeds, then the text should be wrapped in lines if 
the text is long. 

The ordering of items in unit and refresh relations cause or effect lists are not used for display 
purpose. If a WRITE_AS_ONE relation makes it necessary to complete the displayed items 
then these additional items shall appear in the order of the items in the WRITE_AS_ONE 
relation. 

4.5.2 Layout rules for WIDTH and HEIGHT 

VARIABLE, CHART, GRAPH and GRID have the attributes WIDTH and HEIGHT. The WIDTH 
attribute defines the number of columns of display. The HEIGHT attribute defines the vertical 
size of the display as a multiple of the minimum height for a standard input field, see Table 6. 
The default value of WIDTH and HEIGHT is MEDIUM for CHART, GRAPH, GRID;  
XXX_SMALL for VARIABLE height, and width. 

The width of the EDD application should be split in up to a maximum of 5 columns depending 
on the number of COLUMNBREAKs and the WIDTH of the containers. Columns that would 
span to a column higher than the 5th column shall be moved to the next row through an 
implied ROWBREAK. 

An implied ROWBREAK has the same behaviour as an explicitly defined ROWBREAK. 

The WIDTH and HEIGHT attribute of a container defines how many columns the container 
spans and the height in multiples of minimum height possible for a simple field e.g. variable, 
constance string, menu and method reference, see Table 6. 

Table 6 – WIDTH and HEIGHT span and applicability 

WIDTH and HEIGHT Value WIDTH HEIGHT Applicability 

XXX_SMALL 1 1 VARIABLE 

XX_SMALL 1 3 CHART 

GRAPH 

GRID 

VARIABLE 

X_SMALL 1 5 CHART 

GRAPH 

GRID 

VARIABLE 

SMALL 2 7 CHART 

GRAPH 

GRID 

VARIABLE 

MEDIUM 3 8 CHART 

GRAPH 

GRID 

VARIABLE 

LARGE 4 9 CHART 

GRAPH 

GRID 

VARIABLE 

X_LARGE 5 11 CHART 

GRAPH 

GRID 

VARIABLE 

XX_LARGE 5 13 CHART 

GRAPH 

GRID 

VARIABLE 

 
4.5.3 Layout rules for COLUMNBREAK and ROWBREAK 

4.5.3.1 Layout for protruding elements 

The EDD application shall insert all the menu items in a single column one after the other until 
it encounters a COLUMNBREAK. The COLUMNBREAK will cause the next menu item to be 
inserted in the highest row in the next column. If there is a menu item occupying a region of 
space in the next column, then the menu item will be inserted in the row immediately below 
the occupying menu item. See Figure 12 for EDD source code example and Figure 13 for the 
corresponding layout. 

MENU protruding_elements 
{ 
 LABEL "protruding_elements"; 
 STYLE WINDOW; 
 ITEMS  
 { 
  Element1, 
  Element2, 
  Element3, 
  Element4, 
  SmallGraph5,     //WIDTH SMALL (2 columns), HEIGHT SMALL (7 heigth units) 
  Element6, 
  Element7, 
  Element8, 
  COLUMNBREAK, 
  Element9, 
  Element10, 
  Element11, 
  COLUMNREAK, 
  Element12, 
  Element13, 
  Element14, 
 } 
} 

Figure 12 – EDD source code for layout for protruding elements example 

 

CB CB
 

Figure 13 – Layout for protruding elements 

4.5.3.2 Layout for partially filled rows 

The EDD application shall insert all the menu items in a single column one after the other until 
it encounters a ROWBREAK. A ROWBREAK will cause the next menu item to be inserted in 
the lowest row in the 1st column (it acts as a carriage return). The ROWBREAK shall also 
produce a horizontal barrier so that any further COLUMNBREAK will prevent a menu item 
from being inserted above this line. See Figure 14 for EDD source code example and 
Figure 15 for the corresponding layout. 

MENU Fig13 
{ 
  LABEL "Fig13"; 
 STYLE WINDOW; 
 ITEMS 
 { 
  Element1, 
  Element2, 
  Element3, 
  COLUMNBREAK, 
  Element4, 
  Element5, 
  Element6, 
  Element7, 
  ROWBREAK, 
  SmallGraph8, //WIDTH SMALL, HEIGHT SMALL, occupies 2 columns 
  Element9, 
  Element10, 
  Element11, 
  COLUMNBREAK, 
  Element12, 
  Element13, 
  Element14, 
              COLUMNREAK, 
  Element15, 
  Element16, 
  Element17, 
 } 
} 

Figure 14 – EDD source code for layout for partially filled rows example 

 

CB

CB

RB

CB
 

Figure 15 – Layout for partially filled rows 

Figure 17 shows that menu items in the rows with two columns (rows 1 and 3) have expanded 
cell space available such that the row takes up the same width as the rows with three 
columns. Figure 16 shows the corresponding EDD source code example. 

MENU partially_filled_rows 
{ 
  LABEL "partially filled rows"; 
 STYLE WINDOW; 
 ITEMS 
 { 
  Element1, 
  Element2, 
  COLUMNBREAK, 
  Element3, 
  Element4, 
  ROWREAK, 
  Element5, 
  Element6, 
  COLUMNBREAK, 
  Element7, 
  Element8, 
  COLUMNBREAK, 
  Element9, 
  Element10, 
  ROWBREAK, 
  Element11, 
  Element12, 
              COLUMNBREAK, 
  Element13, 
  Element14 
 } 
} 

Figure 16 – EDD source code for layout for partially filled rows example 

 

Figure 17 – Layout for partially filled rows 

4.5.3.3 Layout for oversized elements 

The EDD application shall insert all the menu items in a single column one after the other until 
it encounters a COLUMNBREAK. The COLUMNBREAK will cause the next menu item to be 
inserted in the highest row in the next column. If the menu item to be inserted is wider than 
the number of columns available, then the menu item will be inserted in the lowest row in the 
1st column (i.e., implied ROWBREAK). See Figure 18 for EDD source code example and 
Figure 19 for the corresponding layout. 

MENU layout_for_oversized_elements 
{ 
  LABEL "layout for oversized elements"; 
 STYLE WINDOW; 
 ITEMS 
 { 
  Element1, 
  Element2, 
  Element3, 
  COLUMNBREAK, 
  Element4, 
  Element5, 
  Element6, 
  COLUMNBREAK, 
  Element7, 
  Element8, 
  Element9, 
  COLUMNBREAK, //Results in implied ROWBREAK 
  MediumGraph10 //WIDTH MEDIUM, HEIGHT MEDIUM, occupies 3 columns 
 } 
} 

Figure 18 – EDD source code for layout for oversized elements example 

 

CB CB

Implied RB

CB

 

Figure 19 – Layout for oversized elements 

4.5.3.4 Layout for columns in stacked group 

The EDD application shall insert all the menu items in a single column one after the other until 
it encounters a COLUMNBREAK. The COLUMNBREAK will cause the next menu item to be 
inserted in the highest row in the next column. If the menu item resides in a nested menu 
(such as a GROUP), then the item will remain inside the parent menu. See Figure 20 for EDD 
source code example and Figure 21 for the corresponding layout. 

MENU layout_for_columns_in_stacked_group 
{ 
  LABEL "layout for columns in stacked group"; 
 STYLE WINDOW; 
 ITEMS 
 { 
  Element1, 
  Element2, 
  Element3, 
  Element4, 
  GroupWithThreeColumns5, 
  COLUMNBREAK, 
  COLUMNBREAK, 
  COLUMNBREAK, 
  Element6, 
  Element7, 
  Element8, 
  Element9, 
  Element10, 
  Element11, 
  Element12, 
  Element13 
 } 
} 

Figure 20 – EDD source code example for a layout for columns in stacked group 

Group

CB CB

CB CB CB

 

Figure 21 – Layout for columns in stacked group 

Graphs inside groups shall be handled equivalently as multiple columns inside nested group. 
See Figure 22 for EDD source code example and Figure 23 for the corresponding layout. 

MENU layout_for_columns_in_stacked_group_with_a_graph 
{ 
  LABEL "layout for columns in stacked group with a graph"; 
 STYLE WINDOW; 
 ITEMS 
 { 
  Element1, 
  Element2, 
  Element3, 
  Element4, 
  GroupWithTreeEllementsOneColumnsBreakChartWidthSmall, 
  COLUMNBREAK, 
  COLUMNBREAK, 
  COLUMNBREAK, 
               Element8, 
               Element9, 
               Element10, 
               Element11, 
               Element12, 
               Element13, 
               Element14, 
               Element15 
 } 
} 

Figure 22 – EDD source code for layout for columns 
with GRAPHs in stacked group example 

CB CB CB

CHART / GRAPH (WIDTH SMALL; HEIGHT X_SMALL)Group

CB

 

Figure 23 – Layout for columns with GRAPHs in stacked group 

4.5.4 Layout examples 

4.5.4.1 Single-column layout 

The simplest layout is a list of variables, methods, edit displays, static text, graphs and 
charts. 

MENU overview_window 
{ 
  LABEL "Overview"; 
 STYLE WINDOW; 
 ITEMS 
 { 
  primary_variable,   /* variable */ 
        upper_range_value,   /* variable */ 
        lower_range_value,   /* variable */ 
       master_reset,    /* method */ 
        self_test    /* method */ 
 } 
} 

Figure 24 – Example of an EDD for an overview menu 

The EDD in Figure 24 displays the items vertically in a window. The items are displayed 
starting at the left side of the screen and take up as much of the window as needed, see 
Figure 25.  

Overview

13.48Primary Variable:

20.00Upper Range Value:

10.00Lower Range Value:

Reset...

Test...

psi

psi

psi

 

Figure 25 – Example of an EDD application for an overview window 

4.5.4.2 Multi-column and -row layout 

Multiple columns of variables, methods, groups, images, graphs and charts may also be used, 
see Figure 26. 

MENU overview_page 
{ 
    LABEL "Overview"; 
    STYLE PAGE; 
    ITEMS 
    { 
        upper_range_value,  /* variable */ 
        lower_range_value,  /* variable */ 
        upper_sensor_limit,  /* variable */ 
        lower_sensor_limit,  /* variable */ 
        COLUMNBREAK,   /* column break */ 
        tag,    /* variable */ 
        units,    /* variable */ 
        damping,   /* variable */ 
        transfer_function  /* variable */ 
    } 
} 

Figure 26 – Example of an EDD using COLUMNBREAK 

A COLUMNBREAK in Figure 26 is used to indicate a column break. All of the elements 
preceding the COLUMNBREAK will be placed in one column and all the elements following 
the COLUMNBREAK will be placed into the next column, see Figure 27.  

Overview

20Upper Range Value:

10Lower Range Value:

100Upper Sensor Limit:

0Lower Sensor Limit:

tT-101Tag:

Units:

15Damping:

Transfer Function:  Linear

 psi

psi

psi

psi

psi

sec

 

Figure 27 – Example of an EDD application for an overview window 

A ROWBREAK in Figure 28 is used to place the following menu items beginning on the left 
side, see Figure 29. 

 
MENU overview_page 
{ 
    LABEL "Overview"; 
    STYLE PAGE; 
    ITEMS 
    { 
        tag,    /* variable */ 
        ROWBREAK, 
        range_values,   /* group */ 
        COLUMNBREAK, 
        sensor_limits,   /* group */ 
        ROWBREAK,  
        units,    /* variable */ 
        damping,   /* variable */ 
        transfer_function  /* variable */ 
        COLUMNBREAK 
    } 
} 
 
MENU range_values 
{ 
    LABEL "Range Values"; 
    STYLE GROUP; 
    ITEMS 
    { 
        upper_range_value,  /* variable */ 
        lower_range_value,  /* variable */ 
    } 
} 
 
 
MENU sensor_limits 
{ 
    LABEL "Sensor Limits"; 
    STYLE GROUP; 
    ITEMS 
    { 
        upper_sensor_limit,  /* variable */ 
        lower_sensor_limit,  /* variable */ 
    } 
} 

Figure 28 – EDD example for an overview window 

Figure 29 shows the result of Figure 28 in an EDD application. Blue dashed boxes show the 
available space. 

Overview

Range Values

60Upper: mBar

20Lower: mBar

Sensor Limits

100Upper: mBar

0Lower: mBar

PIC1023TAG:

20Damping: sec

Transfer Function:

Unit: mBar

Iinear

 

Figure 29 – Example of an EDD application for an overview window 

4.5.4.3 In-line graphs and charts 

Graphs and charts may appear within the columns just like variables and methods, see 
Figure 30 and Figure 31. 

MENU overview_page 
{ 
    LABEL "Overview"; 
    STYLE PAGE; 
    ITEMS 
    { 
        primary_variable,  /* variable */ 
        upper_range_value,  /* variable */ 
        lower_range_value,  /* variable */ 
        COLUMNBREAK,   /* column break */ 
        pv_graph   /* graph */ 
    } 
} 

Figure 30 – Example of an EDD for in-line graphs and charts 

Figure 31 – Example of an EDD application for an in-line graph 

If a graph or chart has with a WIDTH attribute that is equal to XX_SMALL, X_SMALL, SMALL 
or MEDIUM, it shall be displayed within the column. When the WIDTH equals LARGE, 
X_LARGE, or XX_LARGE the description in 4.5.4.4 shall apply.  

4.5.4.4 Full-width graphs and charts 

Graphs and charts may also span multiple columns, see Figure 32 and Figure 33. 

GRAPH pv_graph 
{ 
    WIDTH              MEDIUM; 
    HEIGHT             MEDIUM; 
    … 
} 
 
MENU overview_page 
{ 
    LABEL "Overview"; 
    STYLE PAGE; 
    ITEMS 
    { 
        primary_variable,  /* variable */ 
        COLUMNBREAK,   /* column break */ 
        upper_range_value,  /* variable */ 
        COLUMNBREAK,   /* column break */ 
        lower_range_value,  /* variable */ 
        pv_graph,   /* graph */ 
        reload_pv_graph,  /* method */ 
        COLUMNBREAK,   /* column break */ 
        save_pv_graph   /* method */ 
    } 
} 

Figure 32 – Example of an EDD for full-width graphs and charts 

Figure 33 – Example of an EDD application for a full-width graph 

If a graph or chart has a WIDTH attribute that is equal to X_LARGE, or XX_LARGE, it will 
span the width of the window, dialog, page or group. There may be additional elements before 
and after the graph, in which case the elements before the graph or chart are broken into one 
set of columns in a separate row and the elements following the graph or chart are broken 
into another set of columns in a separate row. 

NOTE There are three columns prior to the graph and two columns following the graph. 

4.5.4.5 Nested containers 

Containers can be nested within other containers. Figure 34 and Figure 35 show an example 
of a page that is nested within a window and of two groups that are nested within the page. 

MENU overview_window 
{ 
    LABEL "Device Overview"; 
    STYLE WINDOW; 
    ITEMS 
    { 
        overview_page        /* page */ 
    } 
} 
MENU overview_page 
{ 
    LABEL "Overview"; 
    STYLE PAGE; 
    ITEMS 
    { 
        range_values,       /* group */ 
        sensor_limits,    /* group */ 
        COLUMNBREAK,    /* column break */ 
        tag,     /* variable */ 
        units,     /* variable */ 
        damping,    /* variable */ 
        transfer_function   /* variable */ 
    } 
} 
MENU range_values 
{ 
    LABEL "Range Values"; 
    STYLE GROUP; 
    ITEMS 
    { 
        upper_range_value,    /* variable */ 
        lower_range_value    /* variable */ 
    } 
} 
MENU sensor_limits 
{ 
    LABEL "Sensor Limits"; 
    STYLE GROUP; 
    ITEMS 
    { 
       upper_sensor_limit,    /* variable */ 
       lower_sensor_limit    /* variable */ 
    } 
} 

Figure 34 – Example of an EDD for nested containers 

Figure 35 – Example of an EDD application for nested containers 

4.5.4.6 Edit displays 

EDIT_DISPLAYs can be referenced in containers. When an edit display appears in a 
container, it shall be represented as a button or hyperlink. When the button is activated, a 
dialog that displays the contents of the edit display shall appear, see Figure 36 and Figure 37. 
The OK and Cancel buttons are not in scope of this specification. 

MENU overview_window 
{ 
    LABEL "Device Overview"; 
    STYLE WINDOW; 
    ITEMS 
    { 
        overview_page       /* page */ 
    } 
} 
MENU overview_page 
{ 
    LABEL "Overview"; 
    STYLE PAGE; 
    ITEMS 
    { 
        tag,     /* variable */ 
        units,     /* variable */ 
        damping,    /* variable */ 
        transfer_function,   /* variable */ 
        range_values    /* edit display */ 
    } 
} 
EDIT_DISPLAY range_values 
{ 
    LABEL "Range Values"; 
    DISPLAY_ITEMS 
    { 
        upper_sensor_limit,   /* variable */ 
        lower_sensor_limit   /* variable */ 
    } 
    EDIT_ITEMS 
    { 
        upper_range_value,   /* variable */ 
        lower_range_value,   /* variable */ 
        units     /* variable */ 
    } 
} 

Figure 36 – Example of an EDD for EDIT_DISPLAYS 

Figure 37 – Example of an EDD application for EDIT_DISPLAYS 

4.5.4.7 Images 

Images can also be placed in containers. If an image is referenced with a menu item INLINE 
qualifier, the image is displayed within the current column of the container, see Figure 38 and 
Figure 39. Otherwise, the image is displayed across the width of the container.   

MENU overview_page 
{ 
    LABEL "Overview";   
    STYLE PAGE; 
    ITEMS 
    { 
        voltage,  /* Reference to an image */ 
        ramp_start,  /* variable */ 
        ramp_end,  /* variable */ 
        COLUMNBREAK, 
        logo(INLINE)  /* Reference to an image which is displayed inline */ 
    } 
} 

Figure 38 – Example of an EDD for images 

Figure 39 – Example of an EDD application for images 

4.5.5 Conditional user interface 

4.5.5.1 Overview 

To control the appearance of user interface elements, the EDDL has different possibilities: 

• the attribute VISIBILITY; 

• conditional expressions on the MENU ITEM; 

• the attribute VALIDITY. 

4.5.5.2 VISIBILITY 

The attribute VISIBILITY of the basic constructs (e.g. MENU, VARIABLE, IMAGE, GRAPH, 
CHART, GRID) allows controlling the appearance of user interface elements. The values 
TRUE or FALSE of VISIBILITY are meant to be evaluated dynamically depending on 
VARIABLE values. 

In case of a MENU, VISIBILITY of the child MENU does not affect the screen layout of the 
MENU. In the screen layout of the MENU, the same space is reserved as if the child MENU is 
shown. 

4.5.5.3 Conditional expressions in MENU ITEM 

Conditional expressions in the MENU ITEM list allow controlling the screen layout. In the 
screen layout, no space is reserved if an element is not shown. 

4.5.5.4 VALIDITY 

The attribute VALIDITY of the basic constructs (e.g. MENU, VARIABLE, IMAGE, GRAPH, 
CHART, GRID) allows controlling the appearance of user interface elements. The values 
TRUE or FALSE of VALIDITY are meant to be evaluated dynamically depending on 
VARIABLE values. 

VALIDITY does affect the screen layout. In the screen layout, no space is reserved if the 
element is not valid. 

NOTE Communication is influenced by VALIDITY, see Clause 10. 

4.6 Graphical elements 

4.6.1 Overview 

Smart microprocessor-based field devices continue to get more sophisticated and complex. In 
some cases, measurements or controllers that used to be impractical or required many pieces 
of equipment have been incorporated into a single device. EDDL supports these very 
sophisticated devices.  

Of course, these constructs can be used in many other types of devices. The EDD developer 
has complete control over the content of the EDD and ultimately decides what EDDL 
constructs are to be used. 

To address the needs of these complex devices, support for the following capabilities was 
added: 

• graphing; 

• charting; 

• pictorial content;  

• tabular data. 

EDDL supports two kinds of graphical data visualizations. These are GRAPH and CHART. A 
CHART is used to display actual values and a GRAPH is used to display stored data that may 
be read from the device or persistent storage. One of the common uses is to compare a 
waveform from the device to a reference waveform or waveform from a specific 
date/operation, which is stored by the EDD application. The main difference between both 
constructs is that the EDD application handles the data of a CHART, and for a GRAPH the 
EDD itself defines, reads and updates the data. Methods for data handling are optional for 
CHARTs but they are typically needed for GRAPHs.  

A GRAPH contains a list of WAVEFORMs that are plotted together. The GRAPH may be 
referenced from a MENU or from within a METHOD. WAVEFORMs have a minimal number of 
mandatory attributes in order to allow a graph to be easily produced by the EDD developer.  
For those wanting more control, a wide range of optional attributes is available (display size, 
axis, key points, etc.). 

These constructs are particularly useful for plotting valve signatures and radar signals. Both 
data from the field device and data stored by the EDDL application may be plotted on the 
same GRAPH. For example, data from the field device can be specified in one WAVEFORM 
and persistent data from a FILE in another WAVEFORM. The two WAVEFORMs can be 
plotted on the same GRAPH. 

While a GRAPH is a snapshot of the device or process properties, a CHART provides a view 
of a time varying trend. A CHART has SOURCEs that are sampled periodically. STRIP, 
Sweep and SCOPE style plots are supported so as to show trends over time. Horizontal, 
vertical, and gauge types allow graphical presentation of a single instantaneous value. 

For some devices, the condition and performance cannot be determined empirically. For these 
devices, the relative performance is the indicator of the condition of the device. The FILE 
construct allows access to device data stored by the EDDL application. The EDD specifies the 
data, which is to be stored with a similar syntax to COLLECTION. The EDD developer defines 
the data to be stored in any combination of VARIABLEs, ARRAYs, COLLECTIONs, or LISTs.  

The LIST construct was added to improve language flexibility when specifying persistent data 
stored in a FILE. The LIST is a variable length array. Data can be added to or removed from 
the list over time. Each entry in the LIST is of the same type as specified by the EDD 
developer. 

The COLLECTION construct supports any combination of member types. Previously, 
collections were only allowed to contain references of one type (VARIABLE, ARRAY, 
COLLECTION, MENU). 

Subclause 4.6 provides guidance and expectations for the use and support of EDDL 
constructs for graphical representations. 

4.6.2 Graph and chart 

4.6.3 Common attributes 

4.6.3.1 Sizes 

The HEIGHT and WIDTH attributes define the size of the CHART and GRAPH relative to the 
window rendered by the EDD application. HEIGHT and WIDTH do not provide absolute values 
for the graph window, but rather provide a range of values from XX_SMALL to XX_LARGE. 
The EDD application is free to establish the actual graph window size. The EDD application 
should render CHARTs and GRAPHs that refer to the same HEIGHT or WIDTH value of the 
same proportion. The default setting for HEIGHT and WIDTH is MEDIUM, see Figure 40. The 
EDD application should render CHARTs and GRAPHs that have the same HEIGHT or WIDTH 
with the same proportion. 

Graph1

Graph2

HEIGHT MEDIUM

HEIGHT SMALL

WIDTH MEDIUM

 

Figure 40 – HEIGHT and WIDTH attributes for CHART and GRAPH 

4.6.3.2 Line characteristics 

The LINE_TYPE attribute defines categories for display attributes for WAVEFORMs that are 
rendered on GRAPHs and SOURCEs that are rendered on CHARTs. Such display attributes 
may include colour and line styles (dash, dotted, etc.). The EDD application may provide user- 
configurable attributes for each LINE_TYPE. WAVEFORMs and SOURCEs with the same 
LINE_TYPE attribute should be rendered with the characteristics. Categories exist for data 
that are classified with DATA 1 to DATA 9, limits and TRANSPARENT, which only provides 
the display with the values and does not provide the connecting line. 

The EMPHASIS attribute is used to differentiate between one or more SOURCEs or 
WAVEFORMs on a given CHART or GRAPH. By default, the WAVEFORM or SOURCE with 
the EMPHASIS attribute that is set to TRUE should be displayed with a larger weight (see 
Figure 41). The EDD application may provide user configurable attributes for EMPHASIS. 

E

EMPHASIS FALSE

EMPHASIS TRUE

 

Figure 41 – EMPHASIS attribute to differentiate one or more SOURCEs or WAVEFORMs 

4.6.4 CHART 

4.6.4.1 Overview 

A CHART is used to display continuous data values, as opposed to a GRAPH, which is used 
to display a dataset. The chart supports multiple data sources and multiple Y-axes. Methods 
can be defined for data retrieval, scrolling and zooming support.  

The background colour of a CHART display is determined by the EDD application. To provide 
a common look and feel, it is recommended that the background colour be white. It is 
recommended that an EDD application support at least six curves to be displayed simulta-
neously. 

The elements of CHART are SOURCE (this defines the source of data), visualization 
elements, chart type and actions. 

4.6.4.2 Chart types 

4.6.4.2.1 Overview 

A chart can be shown in different ways. It can be shown as a horizontal or vertical bar chart, 
as a chart with continuously updated waveforms or as a gauge. If the type is not defined, the 
default is STRIP. 

4.6.4.2.2 GAUGE 

The actual visual element of a gauge is decided by the EDD application. A CHART with a type 
GAUGE can have only one data source. 

4.6.4.2.3 HORIZONTAL_BAR 

The actual format (visual elements) of the display is decided by the EDD application. This 
type dictates a horizontal bar graph type of display. This type of display can have multiple bar 
charts (SOURCE). A bar chart is displayed for each variable. 

4.6.4.2.4 SCOPE 

In SCOPE, when the source values reach the end of the display area of the CHART, the 
display area is erased. The new source values are once again displayed, starting at the 
beginning of the display area. This type of display can have multiple SOURCE definitions. 

